



Performing Two-Way Lookups Using MATCH and INDEX

BY BILL JELEN

Most accountants are comfortable using Excel's VLOOKUP function to look up a value in a table, but performing a two-way lookup (the value where a row and column intersect) is more complex. The solution involves a quirky, flexible, but seemingly useless function: MATCH.

How VLOOKUP Works

VLOOKUP works well for a simple lookup. In Figure 1, for example, a ticker symbol is entered in cell C1, and the VLOOKUP formula in cell D1 returns the company name associated with that symbol:

```
=VLOOKUP(C1,A10:B39,2,FALSE)
```

This formula tells Excel to look at the first column of the lookup range and find a match to the value in cell C1. When the match is found, the formula returns the value in the second column of the lookup range (the "2" in the above formula) from that row of the table. The FALSE ensures that Excel returns exact matches only—no close matches.

The MATCH Function

Now let's assume that you want to

find the corresponding value for a particular stock during a certain month. To do so requires finding the proper symbol from column A and the proper month from row 9, then returning the value from the intersection of that row and column. This is where the MATCH function helps.

The MATCH function is similar to VLOOKUP. Like VLOOKUP, it will look for a value within a range of data. The range of data could be the list of ticker symbols found in A10:A39, or it could be a horizontal range like the list of months found in C9:M9. This means that MATCH essentially can search data sets for which you would normally use VLOOKUP or HLOOKUP. So far, so good.

Like VLOOKUP, the final argument controls if you get an exact match. Although Excel treats FALSE

and the number 0 as equivalent, for some reason, Excel's Help suggests using 0 as the final argument in MATCH to ensure an exact match. If you use 1 as the final argument, Excel will find the closest value that is less than the lookup value (similar to using TRUE in VLOOKUP). If you use -1 as the final argument, Excel will find the closest value that is greater than the lookup value (this has no equivalent in VLOOKUP). Just as in VLOOKUP, the lookup table doesn't need to be sorted when you require an exact match.

Unlike VLOOKUP, the MATCH function doesn't require you to specify which column to return from the table. Consequently, while VLOOKUP has four arguments, you omit the third argument when using MATCH.

Here is what stops most people: Instead of returning a value from a table, the MATCH function tells you the relative row number or column number in which the match is found! "Which row number? What good is that? Have you ever had a manager call and ask to know which row something is found in?" When I originally read about MATCH, I could never figure out why I would care to use this function.



Figure 1

	A	B	C	D	E	F	G	H	I	J	K	L	M
1		Enter a symbol:	T	AT&T INC.		=VLOOKUP(C1,\$A\$10:\$M\$39,2,FALSE)							
2		Enter a month:	Oct-07										
3		Row where symbol is found:	5	=MATCH(C1,A10:A39,0)									
4		Column where month found:	4	=MATCH(C2,C9:M9,0)									
5		Value at intersection:	0.355	=INDEX(C10:M39,C3,C4)									
6		Single formula:	0.355	=INDEX(C10:M39,MATCH(C1,A10:A39,0),MATCH(C2,C9:M9,0))									
7		Alternate formula:	0.355	=VLOOKUP(C1,\$A\$10:\$M\$39,MATCH(C2,A9:M9,0),FALSE)									
8													
9	Symbol	Company	Jul-07	Aug-07	Sep-07	Oct-07	Nov-07	Dec-07	Jan-08	Feb-08	Mar-08	Apr-08	May-08
10	MMM	3M COMPANY	0.000	0.480	0.000	0.000	0.480	0.000	0.000	0.500	0.000	0.000	0.500
11	AA	ALCOA INC	0.000	0.170	0.000	0.170	0.000	0.000	0.000	0.170	0.000	0.170	0.000
12	AXP	AMER EXPRESS INC	0.150	0.000	0.000	0.150	0.000	0.000	0.180	0.000	0.000	0.180	0.000
13	AIG	AMER INTL GROUP INC	0.000	0.000	0.200	0.000	0.000	0.200	0.000	0.000	0.200	0.000	0.000
14	T	AT&T INC.	0.355	0.000	0.000	0.355	0.000	0.000	0.400	0.000	0.000	0.400	0.000
15	BAC	BK OF AMERICA CP	0.000	0.000	0.640	0.000	0.000	0.640	0.000	0.000	0.640	0.000	0.000
16	BA	BOEING CO	0.000	0.400	0.000	0.000	0.400	0.000	0.000	0.400	0.000	0.000	0.400
17	CAT	CATERPILLAR INC	0.360	0.000	0.000	0.360	0.000	0.000	0.360	0.000	0.000	0.360	0.000

Examples of MATCH

In cell C3, a MATCH function tells Excel to look for the stock symbol from C1 within the range of A9:A39. When a match is found, Excel will return “5” to indicate that the symbol is found in the fifth row of A9:A39, which is cell A14.

The formula in C4 shows how Excel can be used to search a horizontal row of headings. The formula tells Excel to search for the month from C2 within the range of C9:M9. The result of “4” indicates that October 2007 is the fourth heading within C9:M9.

Bringing It Together with INDEX

To make the MATCH results useful, use another function called INDEX. There are two variations of this function, but here you would use this version:

=INDEX(Rectangular Range,
Which Row, Which Column)

Excel will look through the rectangular range and return the value at the intersection of the row and column numbers passed to the function.

In the figure, the formula in cell C5, =INDEX(C10:M39,C3,C4), will return the value at the intersection of a particular symbol and a particular column. Note that the row and column numbers refer to the relative row within the table. Asking for row 5 of a range that starts in C10 will return a value from row 14.

Of course, you don’t need to enter the formula in three cells. You could easily use a MATCH function as the second and third arguments of the formula:

=INDEX(C10:M39,MATCH(C1,A10:A39,0),MATCH(C2,C9:M9,0))

Or if you want to mix a VLOOKUP and a MATCH, you can use the MATCH function as the third argu-

ment of the VLOOKUP. Note that the lookup range has been adjusted to include columns A and B in the MATCH:

=VLOOKUP(C1,\$A\$10:\$M\$39,
MATCH(C2,A9:M9,0),FALSE)

To master the two-way lookup, develop an understanding of the MATCH function. It works very similar to the VLOOKUP that you probably already use all the time. While the MATCH function does the majority of the work in these formulas, it’s the INDEX function that allows Excel to find the intersection between a particular row and column. ■

Bill Jelen is the host of MrExcel.com. To bring his Power Excel seminar to your next chapter PD day, visit www.MrExcel.com/speaking.html. Send questions for future articles to IMA@MrExcel.com.